

## SAS code for “Location-scale models for multilevel ordinal data: between- and within-subjects variance modeling”

Donald Hedeker, Michael Berbaum, & Robin Mermelstein

*Journal of Probability and Statistical Science*, 2006, volume 4, pages 1-20.

SAS PROC NLMIXED can be used to perform the analyses presented in this article. Syntax is presented below for several of the models, where uppercase lettering identifies SAS syntax and lowercase letters are for user-defined variables or parameters.

In this dataset, the variable `id` is a subject identifier, and the ordinal outcome variable `cphysc` takes on values -2, -1, 0, 1, 2. The variables `smkb` and `smkc` are indicator variables designating the MID and HI smoking history groups, respectively. The first example is for Model Ia, the mixed-effects proportional odds model.

### Model Ia

```
PROC NLMIXED;
PARMS b1=0 b2=0 sd=1 i1=-2 i2=-1 i3=1 i4=2;
z1 = i1 - (b1*smkb + b2*smkc + sd*u);
z2 = i2 - (b1*smkb + b2*smkc + sd*u);
z3 = i3 - (b1*smkb + b2*smkc + sd*u);
z4 = i4 - (b1*smkb + b2*smkc + sd*u);
IF (cphysc=-2) THEN p = 1 / (1 + EXP(-z1));
ELSE IF (cphysc=-1) THEN p = (1/(1 + EXP(-z2))) - (1/(1 + EXP(-z1)));
ELSE IF (cphysc=0) THEN p = (1/(1 + EXP(-z3))) - (1/(1 + EXP(-z2)));
ELSE IF (cphysc=1) THEN p = (1/(1 + EXP(-z4))) - (1/(1 + EXP(-z3)));
ELSE IF (cphysc=2) THEN p = 1 - (1 / (1 + EXP(-z4)));
l1 = LOG(p);
MODEL cphysc ~ GENERAL(l1);
RANDOM u ~ NORMAL(0,1) SUBJECT=id;
RUN;
```

Next, consider the syntax below for Model Ib, which introduces a scaling parameter across the three smoking history groups. The variable `smklev` is coded -1, 0, 1 for the three groups, from LO to MID to HI, respectively. In what follows, we only list the syntax statements that change relative to the syntax listed above for Model 1a.

#### Model Ib

```
PARMS b1=0 b2=0 sd=1 i1=-2 i2=-1 i3=1 i4=2 t1=0;
z1 = (i1 - (b1*smkb + b2*smkc + sd*u)) / EXP(smkleve*t1);
z2 = (i2 - (b1*smkb + b2*smkc + sd*u)) / EXP(smkleve*t1);
z3 = (i3 - (b1*smkb + b2*smkc + sd*u)) / EXP(smkleve*t1);
z4 = (i4 - (b1*smkb + b2*smkc + sd*u)) / EXP(smkleve*t1);
```

Model IIb additionally adds a linear effect of the smoking history variable, `smklev`, on between-subjects variation.

#### Model IIb

```
PARMS b1=0 b2=0 sd1=1 sd2=.5 i1=-2 i2=-1 i3=1 i4=2 t1=0;
z1 = (i1 - (b1*smkb + b2*smkc + sd1*u + smkleve*sd2*u)) / EXP(smkleve*t1);
z2 = (i2 - (b1*smkb + b2*smkc + sd1*u + smkleve*sd2*u)) / EXP(smkleve*t1);
z3 = (i3 - (b1*smkb + b2*smkc + sd1*u + smkleve*sd2*u)) / EXP(smkleve*t1);
z4 = (i4 - (b1*smkb + b2*smkc + sd1*u + smkleve*sd2*u)) / EXP(smkleve*t1);
```

Model IIIc has separate smoking history group effects on both between- and within-subjects variation. The variable `smka` is an indicator variable for the LO smoking history group.

#### Model IIIc

```

PARMS b1=0 b2=0 sd1=1 sd2=1 sd3=1 i1=-2 i2=-1 i3=1 i4=2 t1=0 t2=0;
z1 = (i1 - (b1*smkb + b2*smkc + smka*sd1*u + smkb*sd2*u + smkc*sd3*u))
      / EXP(smkb*t1 + smkc*t2);
z2 = (i2 - (b1*smkb + b2*smkc + smka*sd1*u + smkb*sd2*u + smkc*sd3*u))
      / EXP(smkb*t1 + smkc*t2);
z3 = (i3 - (b1*smkb + b2*smkc + smka*sd1*u + smkb*sd2*u + smkc*sd3*u))
      / EXP(smkb*t1 + smkc*t2);
z4 = (i4 - (b1*smkb + b2*smkc + smka*sd1*u + smkb*sd2*u + smkc*sd3*u))
      / EXP(smkb*t1 + smkc*t2);

```

Finally, the syntax for model IVa indicates how the effects of the covariates `smkb` and `smkc` can vary across the cumulative logits.

#### Model IVa

```

PARMS a1b=0 a1c=0 a2b=0 a2c=0 a3b=0 a3c=0 a4b=0 a4c=0 sd=1 i1=-2 i2=-1 i3=1 i4=2;
z1 = i1 - (a1b*smkb + a1c*smkc + sd*u);
z2 = i2 - (a2b*smkb + a2c*smkc + sd*u);
z3 = i3 - (a3b*smkb + a3c*smkc + sd*u);
z4 = i4 - (a4b*smkb + a4c*smkc + sd*u);

```